

NAG Library Routine Document

G01GEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G01GEF returns the probability associated with the lower tail of the noncentral beta distribution, via the routine name.

2 Specification

```
FUNCTION G01GEF (X, A, B, RLAMDA, TOL, MAXIT, IFAIL)
REAL (KIND=nag_wp) G01GEF
INTEGER MAXIT, IFAIL
REAL (KIND=nag_wp) X, A, B, RLAMDA, TOL
```

3 Description

The lower tail probability for the noncentral beta distribution with parameters a and b and noncentrality parameter λ , $P(B \leq \beta : a, b; \lambda)$, is defined by

$$P(B \leq \beta : a, b; \lambda) = \sum_{j=0}^{\infty} e^{-\lambda/2} \frac{(\lambda/2)^j}{j!} P(B \leq \beta : a, b; 0), \quad (1)$$

where

$$P(B \leq \beta : a, b; 0) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} \int_0^{\beta} B^{a-1} (1-B)^{b-1} dB,$$

which is the central beta probability function or incomplete beta function.

Recurrence relationships given in Abramowitz and Stegun (1972) are used to compute the values of $P(B \leq \beta : a, b; 0)$ for each step of the summation (1).

The algorithm is discussed in Lenth (1987).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Lenth R V (1987) Algorithm AS 226: Computing noncentral beta probabilities *Appl. Statist.* **36** 241–244

5 Parameters

1: X – REAL (KIND=nag_wp) *Input*

On entry: β , the deviate from the beta distribution, for which the probability $P(B \leq \beta : a, b; \lambda)$ is to be found.

Constraint: $0.0 \leq X \leq 1.0$.

2: A – REAL (KIND=nag_wp) *Input*

On entry: a , the first parameter of the required beta distribution.

Constraint: $0.0 < A \leq 10^6$.

- 3: B – REAL (KIND=nag_wp) Input
On entry: b , the second parameter of the required beta distribution.
Constraint: $0.0 < B \leq 10^6$.
- 4: RLAMDA – REAL (KIND=nag_wp) Input
On entry: λ , the noncentrality parameter of the required beta distribution.
Constraint: $0.0 \leq \text{RLAMDA} \leq -2.0 \log(U)$, where U is the safe range parameter as defined by X02AMF.
- 5: TOL – REAL (KIND=nag_wp) Input
On entry: the relative accuracy required by you in the results. If G01GEF is entered with TOL greater than or equal to 1.0 or less than $10 \times \text{machine precision}$ (see X02AJF), then the value of $10 \times \text{machine precision}$ is used instead.
 See Section 7 for the relationship between TOL and MAXIT.
- 6: MAXIT – INTEGER Input
On entry: the maximum number of iterations that the algorithm should use.
 See Section 7 for suggestions as to suitable values for MAXIT for different values of the parameters.
Suggested value: 500.
Constraint: $\text{MAXIT} \geq 1$.
- 7: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if $\text{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G01GEF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

- On entry, $A \leq 0.0$,
- or $A > 10^6$,
- or $B \leq 0.0$,
- or $B > 10^6$,
- or $\text{RLAMDA} < 0.0$,
- or $\text{RLAMDA} > -2.0 \log(U)$, where $U =$ safe range parameter as defined by X02AMF,
- or $X < 0.0$,

or $X > 1.0$,
 or $MAXIT < 1$.

If on exit $IFAIL = 1$ then G01GEF returns zero.

$IFAIL = 2$

The solution has failed to converge in $MAXIT$ iterations. You should try a larger value of $MAXIT$ or TOL . The returned value will be an approximation to the correct value.

$IFAIL = 3$

The probability is too close to 0.0 or 1.0 for the algorithm to be able to calculate the required probability. G01GEF will return 0.0 or 1.0 as appropriate, this should be a reasonable approximation.

$IFAIL = 4$

The required accuracy was not achieved when calculating the initial value of $P(B \leq \beta : a, b; \lambda)$. You should try a larger value of TOL . The returned value will be an approximation to the correct value.

7 Accuracy

Convergence is theoretically guaranteed whenever $P(Y > MAXIT) \leq TOL$ where Y has a Poisson distribution with mean $\lambda/2$. Excessive round-off errors are possible when the number of iterations used is high and TOL is close to *machine precision*. See Lenth (1987) for further comments on the error bound.

8 Further Comments

The central beta probabilities can be obtained by setting $RLAMDA = 0.0$.

9 Example

This example reads values for several beta distributions and calculates and prints the lower tail probabilities until the end of data is reached.

9.1 Program Text

```

Program g01gef

!      G01GEF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: g01gef, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: a, b, prob, rlamda, tol, x
      Integer                     :: ifail, maxit
!      .. Executable Statements ..
      Write (nout,*) 'G01GEF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) '      X      A      B      RLAMDA      PROB'
      Write (nout,*)

```

```

!      Use default tolerance
      tol = 0.0E0_nag_wp
      maxit = 100

d_lp: Do
  Read (nin,*,Iostat=ifail) x, a, b, rlamda
  If (ifail/=0) Then
    Exit d_lp
  End If

!      Calculate probability
      ifail = -1
      prob = g01gef(x,a,b,rlamda,tol,maxit,ifail)
      If (ifail/=0) Then
        If (ifail<3) Then
          Exit d_lp
        End If
      End If

!      Display results
      Write (nout,99999) x, a, b, rlamda, prob
End Do d_lp

99999 Format (1X,4F8.3,F8.4,A,I1)
End Program g01gef

```

9.2 Program Data

G01GEF Example Program Data

0.25	1.0	2.0	1.0	:X A B RLAMDA
0.75	1.5	1.5	0.5	:X A B RLAMDA
0.5	2.0	1.0	0.0	:X A B RLAMDA

9.3 Program Results

G01GEF Example Program Results

X	A	B	RLAMDA	PROB
0.250	1.000	2.000	1.000	0.3168
0.750	1.500	1.500	0.500	0.7705
0.500	2.000	1.000	0.000	0.2500
